
Abstract. Based on an inductive definition of triangulations, a theory of
undirected planar graphs is developed in Isabelle/HOL. The proof of the
5 colour theorem is discussed in some detail, emphasizing the readability
of the computer assisted proofs.

1 Introduction

It is well known that traditional mathematics is in principle reducible to logic.
There are two main motivations for carrying out computer-based formalizations
of mathematics in practice: to demonstrate the actual feasibility (in a particular
domain), and to develop the formal underpinnings for some applications. Our
formalization of graph theory is driven by both motivations.
First of all we would like to explore how feasible computer proofs are in an

area that is (over)loaded with graphic intuition and does not have a strong al-
gebraic theory, namely the realm of planar graphs. A first experiment in that
direction was already reported by Yamamoto et al. [18]. Their main result was
Euler’s formula. This is just one of many stepping stones towards our main result,
the 5 colour theorem (5CT). At the same time we try to lay the foundations for
a project one or two orders of magnitude more ambitious: a computer-assisted
proof of the 4 colour theorem (4CT). We come back to this point in the conclu-
sion. Finally we have an orthogonal aim, namely of demonstrating the advantage
of Wenzel’s extension of Isabelle, called Isar [14], namely readable proofs. Hence
some of the formal proofs are included, together with informal comments. It is
not essential that the reader understands every last detail of the formal proofs.
Isar is as generic as Isabelle but our work is based on the logic HOL.
There is surprisingly little work on formalizing graph theory in the literature.

Apart from the article by Yamamoto et al. on planar graphs [18], from which we
inherit the inductive approach, we are only aware of three other publications [17,
6, 19], none of which deal with planarity.
Before we embark on the technical details, let us briefly recall the history and

main proof idea for the 4 and 5 colour theorem. The 4CT was first conjectured
in 1853 by Francis Guthrie as a map colouring problem. The first incorrect proof
attempts were given 1879 by Kempe and 1880 by Tait. Kempe already used an
argument today known as Kempe chains. The incorrectness of Kempe’s proof was



pointed out by Heawood in 1890, but he could weaken Kempe’s argument to the
5CT. In 1969 Heesch [9] introduced the methods of reducibility and discharging
which were used in the first proof by Appel and Haken in 1976 [1, 2, 3]. The proof
was simplified and improved 1995 by Robertson, Sanders, Seymour and Thomas
[11, 12]. Both proofs rely heavily on large case analyses performed by (unverified)
computer programs, which is contrary to the tradition of mathematical proofs.
The proof of the 5CT is by induction on the size of the graph g (although

many textbooks [15, 8] prefer an indirect argument). Let g be non-empty. It can
be shown that any non-empty planar graph has a vertex of degree ≤ 5. Let v
be such a vertex in g and let g′ be g without v. By induction hypothesis g′ is
5-colourable because g′ is smaller than g. Now we distinguish two cases. If v has
degree less than 5, there are fewer neighbours than colours, and thus there is
at least one colour left for v. Thus g is also 5-colorable. If v has degree 5, it is
more complicated to construct a 5-colouring of g′ from one of g. This core of the
proof is explained in the body of the paper. In principle, the proof of the 4CT
is similar, but immensly more complicated in its details.
We have intentionally not made an effort to find the slickest proof of the 5CT

but worked with a standard one (except for using induction rather than a mini-
mal counterexample). This is because we intended the proof to be a benchmark
for our formalization of graph theory.

1.1 Notation

We briefly list the most important non-standard notations in Isabelle/Isar.
The notation [[P1; . . . ;Pn]] =⇒ P is short for the iterated implication P1 =⇒

. . . =⇒ Pn =⇒ P . The symbol
∧

denotes Isabelle’s universal quantifier from
the meta-logic. Set comprehension is written {x.P} instead of {x | P}. Function
card maps a finite set to its cardinality, a natural number. The notation c ‘ A

is used for the image of a set A under a function c.
The keyword constdefs starts the declaration and definition of new con-

stants; “≡”is definitional equality. Propositions in definitions, lemmas, and proofs
can be preceded by a name, written “name: proposition”; We use the technique
of suppressing subproofs, denoted by the symbol “〈proof 〉”, to illustrate the
structure of a proof. The subproofs may be trivial subproofs, by simplification
or predicate calculus or involve some irrelevant Isabelle specific steps. If a propo-
sition can be proved directly, its proof starts with the keyword by followed by
some builtin proof method like simp, arith, rule, blast or auto [10]; The name
?thesis in an Isar proof refers to the proposition we are proving at that moment.

1.2 Overview

In §2 we introduce graphs and their basic operations. In §3 we give an inductive
definition of near triangulations and define planarity using triangulations. In
§4 we discuss two fundamental topological properies of planar graphs and their
representations in our formalization. In §5 we introduce colourings and Kempe
chains and present the Isar proof of the 5CT.

2



2 Formalization of graphs

2.1 Definition of graphs

In this section we define the type of graphs and introduce some basic graph
operations (insertion and deletion of edges and deletion of vertices) and some
classes of graphs.
A (finite) graph g consists of a finite set of vertices V g and a set of edges

E g , where each edge is a two-element subset of V g. In the field of planar graphs
or graph colouring problems, isolated vertices do not matter much, since we
can always assign them any colour. Hence we consider only vertices which are
connected to another vertex, and leave out an explicit representation of V g.
This leads to an edge point of view of graphs, which keeps our formalisation as
simple as possible. We do not restrict the definition of graphs to simple graphs
but most of the graphs we actually work with are simple by construction.
We define a new type α graph as an isomorphic copy of the set of finite sets

of two-element sets with the morphisms edges and graph-of :

edges

graph-of

α graph α set set

Fig. 1. Definition of α graph

typedef α graph = {g ::α set set . finite g ∧ (∀ e ∈ g . ∃ x y . e = {x , y})}

morphisms edges graph-of

The representing set is not empty as it contains, for example, the empty set.
We introduce the symbol E for the function edges and define V g as the set

of all vertices which have an incident edge.

constdefs vertices :: α graph ⇒ α set (V)
V g ≡

⋃

(E g)

2.2 Basic graph operations

Size We define the size of a graph as the (finite) number of its vertices.

size g ≡ card (V g)

Empty graph, insertion and deletion of edges We introduce the following
basic graph operations: the empty graph ∅, insertion g ⊕ {x , y} of an edge,

deletion g ª {x , y} of an edge and deletion g ¯ x of a vertex x, i.e. deletion of
all edges which are incident with x.

constdefs empty-graph :: α graph (∅)
∅ ≡ graph-of {}

ins :: α graph ⇒ α set ⇒ α graph (infixl ⊕ 60)
e = {x , y} =⇒ g ⊕ e ≡ graph-of (insert e (E g))

del :: α graph ⇒ α set ⇒ α graph (infixl ª 60)
g ª e ≡ graph-of (E g − {e})

del-vertex :: α graph ⇒ α ⇒ α graph (infixl ¯ 60)
g ¯ x ≡ graph-of {e. e ∈ E g ∧ x /∈ e}

Subgraph A graph g is a (spanning) subgraph of h, g ¹ h, iff g can be extended
to h by only inserting edges between existing vertices.

constdefs subgraph :: α graph ⇒ α graph ⇒ bool (infixl ¹ 60)
g ¹ h ≡ V g = V h ∧ E g ⊆ E h

Degree The neighbourhood Γ g x of x is the set of incident vertices. The degree
of a vertex x ∈ V g is the number of neighbours of x.

constdefs neighbours :: α graph ⇒ α ⇒ α set (Γ )
Γ g x ≡ {y . {x , y} ∈ E g}

degree :: α graph ⇒ α ⇒ nat
degree g x ≡ card (Γ g x )

2.3 Some classes of graphs

Complete graphs with n vertices are simple graphs where any two different
vertices are incident. For example a triangle is a complete graph with 3 vertices.

constdefs K 3 :: α ⇒ α ⇒ α ⇒ α graph (K 3)
K 3 a b c ≡ ∅ ⊕ {a, b} ⊕ {a, c} ⊕ {b, c}

Cycles are simple connected graphs where any vertex has degree 2. We do not
require connectivity, and thus obtain sets of cycles.

constdefs cycles :: α graph ⇒ bool

cycles g ≡ simple g ∧ (∀ v ∈ V g . degree g v = 2)

We introduce abbreviations cycle-ins-vertex and cycle-del-vertex for insertion
and deletion of vertices in cycles (see Fig. 2).

constdefs cycle-ins-vertex :: α graph ⇒ α ⇒ α ⇒ α ⇒ α graph
cycle-ins-vertex g x y z ≡ g ª {x , z} ⊕ {x , y} ⊕ {y , z}

cycle-ins-vertex

cycle-del-vertex
y

z

x

Fig. 2. Insertion and deletion of vertices in cycles

cycle-del-vertex :: α graph ⇒ α ⇒ α ⇒ α ⇒ α graph
cycle-del-vertex g x y z ≡ g ª {x , y} ª {y , z} ⊕ {x , z}

3 Triangulations and planarity

3.1 Inductive construction of near triangulations

To lay the foundations for our definition of planarity based on triangulations,
we start in this section with the inductive definition of near triangulations. The
idea of defining planar graphs inductively is due to Yamamoto et al. [18], but
the formulation in terms of near triangulations is due to Wiedijk [16]. We will
then derive some fundamental properties of near triangulations such as Euler’s
theorem and the existence of a vertex of degree at most 5.
Triangulations are plane graphs where any face is a triangle. Near triangu-

lations are plane graphs where any face except the outer face is a triangle. We
define near triangulations g with boundary h and inner faces f in an inductive
way (see Fig. 3): any triangle K 3 a b c of three distinct vertices is a near trian-
gulation. Near triangulations are extended by adding triangles on the outside in
two different ways: in the first case we introduce a new vertex b by inserting two
edges joining b with a and c, where a and c are incident on h. In the other case
we insert a new edge joining a and c, where a and c are both incident with b on
h. The boundary of the new near triangulation is constructed from the original

ins-edgetriangle ins-vertex

a

b

a

b

c

a

b

c
c

Fig. 3. Inductive definition of near triangulations

one by inserting or deleting the vertex b on h. The set of inner faces f is extended
in both cases by the new triangle K 3 a b c.

consts near-triangulations :: (α graph × α graph × α graph set) set
inductive near-triangulations
intros

triangle:
distinct [a, b, c] =⇒ (K 3 a b c, K 3 a b c, {K 3 a b c}) ∈ near-triangulations

ins-vertex :
[[ distinct [a, b, c]; {a, c} ∈ E h; b /∈ V g ;
(g , h, f ) ∈ near-triangulations ]] =⇒
(g ⊕ {a, b} ⊕ {b, c}, cycle-ins-vertex h a b c, f ∪ {K 3 a b c}) ∈ near-triangulations

ins-edge:
[[ distinct [a, b, c]; {a, b} ∈ E h; {b, c} ∈ E h; {a, c} /∈ E g ;
(g , h, f ) ∈ near-triangulations ]] =⇒

(g ⊕ {a, c}, cycle-del-vertex h a b c, f ∪ {K 3 a b c}) ∈ near-triangulations

A near triangulation is a graph g which can be constructed in this way. A
triangulation is a special near triangulation where the boundary consist of 3
vertices. Of course there may be many different ways of constructing a near
triangulation. A boundary H g is some boundary in one possible construction
of g, F g is some set of inner faces in one possible construction of g. Note that
both the inner faces and the boundary (which can be counsidered as the outer
face) are not unique, since some planar graphs can be drawn in different ways.
Note that SOME is Hilbert’s ε-operator, the choice construct in HOL.

constdefs near-triangulation :: α graph ⇒ bool
near-triangulation g ≡ ∃ h f . (g , h, f ) ∈ near-triangulations

triangulation :: α graph ⇒ bool
triangulation g ≡ ∃ h f . (g , h, f ) ∈ near-triangulations ∧ card (V h) = 3

boundary :: α graph ⇒ α graph (H)
H g ≡ SOME h. ∃ f . (g , h, f ) ∈ near-triangulations

faces :: α graph ⇒ α graph set (F)
F g ≡ SOME f . ∃ h. (g , h, f ) ∈ near-triangulations

3.2 Fundamental properties of near triangulations

Our next goal is to prove that any planar graph has at least one vertex of degree
≤ 5. Since our definition of planarity is based on near triangulations, we will see
that it suffices to show that any near triangulation contains a vertex of degree
≤ 5. To this end we will first establish a series of equations about the number
of edges, vertices and faces in near triangulations. Finally we will show that any
near triangulation contains a vertex of degree ≤ 5.
The first equation relates the number of inner faces and the number of edges.

Any face is adjoined by 3 edges, any edge belongs to 2 faces except the edges on
the boundary which belong only to one face. The proof is by induction over the
construction of near triangulations.

lemma nt-edges-faces:

(g , h, f ) ∈ near-triangulations =⇒ 2 ∗ card (E g) = 3 ∗ card f + card (V h) 〈proof 〉

Euler’s theorem for near triangulations is also proved by induction. Note that
F g contains only the inner faces of g, so the statement contains a 1 instead of
the customary 2. We present the top-level structure of the proof, relating the
considered quantities in the extended and original near triangulation. Note that
we introduce an abbreviation ?P for the statement we want to prove.

theorem Euler :
near-triangulation g =⇒ 1 + card (E g) = card (V g) + card (F g)

proof −
assume near-triangulation g
then show 1 + card (E g) = card (V g) + card (F g) (is ?P g (F g))
proof (induct rule: nt-f-induct)
case (triangle a b c)
show ?P (K 3 a b c) {K 3 a b c} 〈proof 〉

next

case (ins-vertex a b c f g h)
have card (E (g ⊕ {a, b} ⊕ {b, c})) = card (E g) + 2 〈proof 〉
moreover have card (V (g ⊕ {a, b} ⊕ {b, c})) = 1 + card (V g) 〈proof 〉
moreover have card (f ∪ {K 3 a b c}) = card f + 1 〈proof 〉
moreover assume ?P g f
ultimately show ?P (g ⊕ {a, b} ⊕ {b, c}) (f ∪ {K 3 a b c}) by simp

next

case (ins-edge a b c f g h)
have card (E (g ⊕ {a, c})) = 1 + card (E g) 〈proof 〉
moreover have card (V (g ⊕ {a, c})) = card (V g) 〈proof 〉
moreover have card (f ∪ {K 3 a b c}) = 1 + card f 〈proof 〉
moreover assume ?P g f
ultimately show ?P (g ⊕ {a, c}) (f ∪ {K 3 a b c}) by simp

qed

qed

The following result is easily achieved by combining the previous ones.

lemma edges-vertices: near-triangulation g =⇒

card (E g) = 3 ∗ card (V g) − 3 − card (V (H g)) 〈proof 〉

The sum of the degrees of all vertices counts the number of neighbours over
all vertices, so every edge is counted twice. The proof is again by induction.

theorem degree-sum: near-triangulation g =⇒

(
∑

v ∈ V g . degree g v) = 2 ∗ card (E g) 〈proof 〉

Now we prove our claim by contradiction. We can complete the proof with a
short calculation combining the two preceding equations.

theorem degree5: near-triangulation g =⇒ ∃ v ∈ V g . degree g v ≤ 5
proof −
assume (∗): near-triangulation g then have (∗∗): 3 ≤ card (V g) by (rule card3)
show ?thesis

proof (rule classical)
assume ¬ (∃ v ∈ V g . degree g v ≤ 5)
then have

∧

v . v ∈ V g =⇒ 6 ≤ degree g v by auto
then have 6 ∗ card (V g) ≤ (

∑

v ∈ V g . degree g v) by simp
also have . . . = 2 ∗ card (E g) by (rule degree-sum)
also from (∗∗) and (∗) have . . . < 6 ∗ card (V g)
by (simp add : edges-vertices) arith

finally show ?thesis by arith
qed

qed

3.3 Planar graphs

In text books, planar graphs are usually defined as those graphs which have
an embedding in the two-dimensional Euclidean plane with no crossing edges.
However, this approach requires a significant amount of topology. We start from
the well-known and easy observation that planar graphs with at least 3 vertices
are exactly those which can be extended to a triangulation by adding only edges.
We do not consider graphs with less than 3 vertices, since a graph with at most
n vertices can always be n-coloured. This leads to the following definition.

constdefs planar :: α graph ⇒ bool

planar g ≡ ∃ g′. triangulation g′ ∧ g ¹ g′

Obviously triangulations are planar. But we also need to know that any near
triangulation g is planar. Of course this is also true, we only have to insert edges
until there are only 3 vertices on the boundary h of g. We can prove this by
induction on the number of vertices in h.

lemma near-triangulation-planar : near-triangulation g =⇒ planar g 〈proof 〉

We have glossed over one issue so far: is the usual topological definition of
triangulation the same as our inductive one? And thus, is our notion of planarity
the same as the topological one? It is clear that all of our inductively generated
near triangulations are near triangulations in the topological sense, and hence
that any planar graph in our sense is also planar in the topological sense. In
the other direction let g be a planar graph in the topological sense such that
g has at least 3 vertices. Then g ¹ g′ for some 2-connected planar graph g′.
By Proposition 4 of [18] g′ can be generated by the inductive definition in [18],
where arbitrary polygons rather than just triangles are added in each step. Now
there is a triangulation g′′ in our sense such that g′ ¹ g′′ — just add edges to the
polygons to subdivide them into triangles. Thus planar g because g ¹ g′′. This
finishes the informal proof that our inductive and the usual topological notion
of planarity coincide.

4 Topological properties of near triangulations

4.1 Crossing paths on near triangulations

This section is dedicated to an obvious topological property of near triangula-
tions shown in Fig. 4 on the left, where the circle is the boundary: any two paths
between v0 and v2 and between v1 and v3 must intersect, i.e. have a vertex in
common. In fact, this is true for all planar graphs. We give an inductive proof
for near triangulations: given two paths in the graph extended by a new triangle,
this induces two sub-paths in the old graph, which must intersect by induction
hypothesis. The cases where the two paths lie completely within the old graph
are trivial by induction hypothesis. Then there are numerous symmetric and
degenerate cases where one of the two paths include (some of) the newly added
edges. In Fig. 4 on the right you see the two main cases, one for inserting a vertex
and one for inserting an edge. In both cases the sub-paths between v0 and a and
between v1 and v3 lie within the old graph and therefore intersect. In the first
case one can also replace the sub-path abc by ac to obtain a path within the old
graph, thus cutting down on the number of case distinctions, which is what we
did in the formal proof.

v1

v2

v3

v0

b c

a a

Fig. 4. Crossing paths

Although it took us a moment to find the pictorial proof, the shock came
when the formal proof grew to 2000 lines. We will not discuss any details but
merely the concepts involved.
A path could either be a graph or a list of vertices. We opted for lists because

of their familiarity and large library. This may have been a mistake because lists
are better suited for directed paths and may require additional case distinctions
in proofs in an undirected setting. Given a graph g, (x ,p,y) ∈ paths g means that
p is a list of distinct vertices leading from x to y. The definition is inductive:

consts paths :: α graph ⇒ (α × α list × α)set
inductive paths g
intros

basis: (x , [x ], x ) ∈ paths g

step: [[ (y , p, z ) ∈ paths g ; {x , y} ∈ E g ; x /∈ set p ]] =⇒ (x , x#p, z ) ∈ paths g

Function set converts a list into a set and the infix # separates the head from
the tail of a list.
From a path p it is easy to recover a set of edges pedges p. Keyword recdef

starts a recursive function definition and measure size is the hint for the auto-
matic termination prover:

consts pedges :: α list ⇒ α set set
recdef measure length
pedges (x#y#zs) = {{x , y}} ∪ pedges (y#zs)

pedges xs = {}

Now we come to the key assumption of our main lemma shown in Fig. 4: the
vertices v0, v1, v2 and v3 must appear in the right order on the boundary. We
express this with the predicate ortho h x y u v where h is the boundary, x must
be opposite y and u opposite v :

constdefs ortho :: α graph ⇒ α ⇒ α ⇒ α ⇒ α ⇒ bool
ortho h x y u v ≡ x 6= y ∧ u 6= v ∧

(∃ p q . (x , p, y) ∈ paths h ∧ (y , q , x ) ∈ paths h ∧ u ∈ set p ∧ v ∈ set q ∧

E h = pedges p ∪ pedges q ∧ set p ∩ set q ⊆ {x , y})

This definition is attractive because it follows very easily that x can be
swapped with y and u with v. But there are many alternatives to this defi-
nition, and in retrospect we wonder if we made the right choice. As it turns out,
almost half the size of this theory, i.e. 1000 lines, is concerned with proofs about
ortho. There is certainly room for more automation. All of theses proofs involve
first-order logic only, no induction.
The main theorem can now be expressed very easily:

theorem crossing-paths: [[ (g , h, f ) ∈ near-triangulations; ortho h v 0 v2 v1 v3;

(v0, p, v2) ∈ paths g ; (v1, q , v3) ∈ paths g ]] =⇒ set p ∩ set q 6= {} 〈proof 〉

We also need to know that we can find 4 orthogonal vertices in a large enough
graph:

lemma cycles-ortho-ex : [[ cycles h; 4 ≤ card (V h) ]] =⇒
∃ v0 v1 v2 v3. ortho h v0 v2 v1 v3 ∧ distinct [v0, v1, v2, v3] 〈proof 〉

4.2 Inversions of triangulations

For the proof of the 5CT we need the following topological property, which is
illustrated in Fig. 5: if we delete a vertex v in a triangulation g, we obtain (in two
steps) a near triangulation with boundary h where the vertices of h are exactly
the neighbours v0, v1, v2, v3, v4 of v.

theorem del-vertex-near-triangulation: [[ triangulation g ; 4 ≤ card (V g) ]] =⇒

∃ h f .(g ¯ v , h, f ) ∈ near-triangulations ∧ V h = Γ g v

a
v4

a

v3

v2 v1

v0

v0

v3

v1

v2

v4

v

Fig. 5. Deletion of a vertex in a triangulation

This property is based on another fundamental but non-trivial property of
triangulations (see Fig. 6). Given a triangulation g, we can consider any face of
g as the boundary of g, i.e. we may invert the given inductive construction of g
such that the boundary is one of the inner faces.

v1

v v0

c b

a

v v0

v1

a

bc

Fig. 6. Inversion of a triangulation

Now we can complete the proof of our previous claim. First we invert the
triangulation such that one of the faces adjoining v is the outer one (first step in
Fig. 5). Now we cut off all edges to neighbours of v and obtain a near triangu-
lation where the vertices on h are exactly the neighbours of v. This is the only
theorem we have not formally proved so far, but it is a topological property and
does not concern colouring. It also is a special case of the following topological
property, which may be seen as a formulation of the Jordan Curve Theorem: if
we delete the inner faces of a near triangulation from a triangulation we again
obtain a near triangulation.

5 The five colour theorem

5.1 Colourings and Kempe chains

A n-colouring of a graph g is a function from V g to {k . k < n} such that any
incident vertices have different colours. A graph g is n-colourable if there is an
n-colouring of g.

constdefs colouring :: nat ⇒ α graph ⇒ (α ⇒ nat) ⇒ bool
colouring n g c ≡
(∀ v . v ∈ V g −→ c v < n) ∧ (∀ v w . {v , w} ∈ E g −→ c v 6= c w)

constdefs colourable :: nat ⇒ α graph ⇒ bool

colourable n g ≡ ∃ c. colouring n g c

A Kempe chain of colours i and j is the subset of all vertices of g which are
connected with v by a path of vertices of colours i and j.
We are able to prove the following major result: we can obtain a new n-

colouring c′ from a n-colouring c by transposing the colours of all vertices in a
Kempe chain.

constdefs kempe :: α graph ⇒ (α ⇒ nat) ⇒ α ⇒ nat ⇒ nat ⇒ α set (K)
K g c v i j ≡ {w . ∃ p. (v , p, w) ∈ paths g ∧ c‘ (set p) ⊆ {i , j}}

constdefs swap-colours :: α set ⇒ nat ⇒ nat ⇒ (α ⇒ nat) ⇒ (α ⇒ nat)
swap-colours K i j c v ≡
if v ∈ K then if c v = i then j else if c v = j then i else c v else c v

lemma swap-colours-kempe: [[ v ∈ V g ; i < n; j < n; colouring n g c ]] =⇒
colouring n g (swap-colours (K g c v i j ) i j c) 〈proof 〉

5.2 Reductions

A important proof principle in the proof of the 5CT is reduction of the problem
of colouring g to the problem of colouring a smaller graph. For the proof of the
5CT we only need two different reductions. One for the case that g has a vertex
v of degree less than 4 and one for the case that g has a vertex v of degree 5.
Both reductions are based on the following property: if a colouring c on g ¯ v

uses less than 5 colours for the neighbours Γ g v of v, then c can be extended to
a colouring on g. We only have to assign to v one of the colours which are not
used in Γ g v.

lemma free-colour-reduction:
[[ card (c ‘ Γ g v) < 5; colouring 5 (g ¯ v) c; triangulation g ]] =⇒

colourable 5 g 〈proof 〉

The reduction for the case that v has degree less than 5 is quite obvious:
there are only 4 neighbours so they cannot use more than 4 colours.

lemma degree4-reduction:

[[ colourable 5 (g ¯ v); degree g v < 5; triangulation g ]] =⇒ colourable 5 g 〈proof 〉

The reduction for the cases that v has degree 5 is a bit more complicated
[8, 7]: we construct a colouring of g from a colouring of g ¯ v where g is a
triangulation and degree g v = 5. We assume g ¯ v is colourable, so we can
choose an (arbitrary) colouring c of g ¯ v. In the case that c uses less than 5
colours in the neighbourhood Γ g v of v, we obviously can extend c on g.

lemma degree5-reduction:
[[ colourable 5 (g ¯ v); degree g v = 5; triangulation g ]] =⇒ colourable 5 g

proof −
assume (∗): triangulation g and degree g v = 5
assume colourable 5 (g ¯ v) then obtain c where colouring 5 (g ¯ v) c ..

have card (c ‘ Γ g v) ≤ 5 〈proof 〉 then show colourable 5 g
proof cases
assume card (c ‘ Γ g v) < 5 show colourable 5 g by (rule free-colour-reduction)

next

In the case that c uses all 5 colours on the neighbours Γ g v of v, we consider
the near triangulation g ¯ v with boundary h on Γ g v. We can find four distinct
orthogonal vertices v0 v1 v2 v3 on h.

assume card (c ‘ Γ g v) = 5 also have (∗∗): card (Γ g v) = 5 〈proof 〉
finally have inj-on c (Γ g v) 〈proof 〉

have 4 ≤ card (V g) 〈proof 〉
with (∗) obtain h f where (g ¯ v , h, f ) ∈ near-triangulations and
V h = Γ g v by (rules dest : del-vertex-near-triangulation)

with (∗∗) have 4 ≤ card (V h) and cycles h 〈proof 〉
then obtain v0 v1 v2 v3 where ortho h v0 v2 v1 v3 and distinct [v0, v1, v2, v3]
by (auto dest : cycles-ortho-ex )

There cannot be a path in g ¯ v in the colours of v 0 and v2 between v0 and v2

and at the same time one in the colours of v 1 and v3 between v1 and v3. The reason
is that the paths would intersect according to the crossing-paths lemma (see Fig. 7 on
the left). This is a contradiction to the fact that all neighbours have different colours.
Thus we can obtain v0

′ and v2
′ such that there is no path in the colours of v 0

′ and v2
′

between v0
′ and v2

′.

v4

v0

v3

v1
v2

v4

v0

v3

v1
v2

Fig. 7. Reduction for the case degree g v = 5

have ¬ ((∃ p. (v0, p, v2) ∈ paths (g ¯ v) ∧ c‘ (set p) ⊆ {c v 0, c v2}) ∧
(∃ p. (v1, p, v3) ∈ paths (g ¯ v) ∧ c‘ (set p) ⊆ {c v 1, c v3}))

proof −
{ fix p assume (v0, p, v2) ∈ paths (g ¯ v) and c‘ (set p) ⊆ {c v 0, c v2}
fix q assume (v1, q , v3) ∈ paths (g ¯ v) and c‘ (set q) ⊆ {c v 1, c v3}
have set p ∩ set q 6= {} by (rule crossing-paths)
then obtain s where s ∈ set p and s ∈ set q by auto
have c s ∈ {c v0, c v2} and c s ∈ {c v1, c v3} 〈proof 〉
moreover have c v0 6= c v1 and c v0 6= c v3 and

c v1 6= c v2 and c v2 6= c v3 〈proof 〉
ultimately have False by auto

} then show ?thesis by auto
qed

then obtain v0
′ v2

′ where v0
′ = v0 ∧ v2

′ = v2 ∨ v0
′ = v1 ∧ v2

′ = v3 and

¬ (∃ p. p ∈ paths (g ¯ v) v0
′ v2

′ ∧ c‘ (set p) ⊆ {c v0
′, c v2

′}) by auto

We consider the Kempe chain K of all vertices which are connected with v 0
′ by

a path in the colours of v0
′ and v2

′. We define a new colouring c′ on g ¯ v by
transposing the colours of v0

′ and v2
′ in K. According to our previous considerations

v2
′ is separated from v0

′, so v2
′ does not lie in K. (See Fig. 7, on the right.) We can

extend the colouring c′ to g by assigning the colour of v0
′ to v.

def K ≡ K (g ¯ v) c v0
′ (c v0

′) (c v2
′)

def c′ ≡ swap-colours K (c v0
′) (c v2

′) c
obtain colouring 5 (g ¯ v) c′ 〈proof 〉

have v2
′ /∈ K 〈proof 〉 then have c′ v2

′ = c v2
′ 〈proof 〉

also have v0
′ ∈ K 〈proof 〉 then have c′ v0

′ = c v2
′ 〈proof 〉

finally have c′ v0
′ = c′ v2

′ .

then have ¬ inj-on c′ (V h) 〈proof 〉 also have V h = Γ g v .
finally have card (c′ ‘ Γ g v) < 5 〈proof 〉
then show colourable 5 g by (rule free-colour-reduction)

qed

qed

5.3 The proof of the five colour theorem

We prove the 5CT using the following rule for induction on size g.

lemma graph-measure-induct :

(
∧

g . (
∧

h. size h < size g =⇒ P h) =⇒ P g) =⇒ P (g ::α graph) 〈proof 〉

We show that any planar graph is 5-colourable by induction on size g (com-
pare [8, 7]). We assume that g is planar. Hence it is the spanning subgraph of a
triangulation t. We know that t contains at least one vertex of degree less than
5. The size of t ¯ v is smaller than the size of g. Moreover, t ¯ v is a near tri-
angulation and therefore planar. It follows from the induction hypothesis (IH )
that t ¯ v must be 5-colourable. Now we construct a colouring of t depending
on a case distinction on the degree of v by applying an appropriate reduction.

theorem 5CT : planar g =⇒ colourable 5 g
proof −
assume planar g then show colourable 5 g
proof (induct rule: graph-measure-induct)
fix g ::α graph
assume IH :

∧

g′::α graph. size g′ < size g =⇒ planar g′ =⇒ colourable 5 g′

assume planar g then obtain t where triangulation t and g ¹ t ..
then obtain v where v ∈ V t and d : degree t v ≤ 5 by (blast dest : degree5)

have size (t ¯ v) < size t .. also have size t = size g ..

also have planar (t ¯ v) ..
ultimately obtain colourable 5 (t ¯ v) by (rules dest : IH )

from d have colourable 5 t
proof cases
assume degree t v < 5 show colourable 5 t by (rule degree4-reduction)

next

assume degree t v = 5 show colourable 5 t by (rule degree5-reduction)
qed

then show colourable 5 g ..
qed

qed

6 Conclusion

Although we have proved what we set out to, the resulting formalization and
proofs are still only a beginning. For a start, the size of the proofs for the crossing-

paths lemma in §4.1 are too large for comfort, and the proof of theorem del-

vertex-neartriangulation still needs to be formalized in Isabelle. It also appears
that both propositions could be obtained as special cases of a discrete version of
the Jordan Curve Theorem. However, in other parts, e.g. Euler’s theorem and
the reduction lemmas, the proofs are roughly in line with what one would expect
from the informal proofs. The complete proof, without the topological properties,
consists of about 20 pages basic graph theory, 15 pages of triangulations, and 7
pages for the 5CT including Kempe chains and reductions.

So what about a full formalization of one of the existing proofs of the 4CT [4,
13]? The top-level proof by Robertson et al. consists of two computer-validated
theorems plus a lemma by Birkhoff [5]. The latter is more complicated than the
5CT, but still reasonably simple. The crucial part are the two computer-validated
theorems. One would either have to verify the correctness of the algorithms, or
let the algorithms produce a trace that can be turned into a proof. Although the
second appears more tractable, in either case one needs a considerable amount of
additional graph theory. Such an undertaking appears within reach of a dedicated
group of experts in theorem proving and graph theory, and its benefits could be
significant.

